POGLAVJE II

II.9 Iskanje lastnih frekvenc z metodo končnih elementov

1. Izpeljava togostne in masne matrike posebnega končnega elementa za prečno nihanje
2. Modeliranje konstrukcije s končnimi elementi
3. Iskanje lastnih frekvenc

1. Izpeljava togostne in masne matrike posebnega končnega elementa za prečno nihanje

V prejšnji vaji je bilo pokazano, da se pri modeliranju konstrukcij s porazdeljeno maso s končnimi elementi, izpeljanimi s statičnimi interpolacijskimi funkcijami, konvergenca rezultatov doseže z povečevanjem števila končnih elementov za opis posameznega konstrukcijskega elementa. Vzrok za ta pojav se skriva v dejstvu, da so dejanski pomiki konstrukcije med lastnim nihanjem opisani s trigonometričnimi funkcijami, ki so funkcije lastne frekvence, pri izpeljavi končnega elementa pa so se uporabile statične interpolacijske funkcije, ki so bili polinomi zgolj tretjega reda.
Ker uporaba trigonometričnih funkcij analizo zaplete, to napelje na misel, da bi se lahko tudi z uporabo polinomov višjega reda dosegli boljši približki trigonometričnim funkcijam, kar bi posledično zahtevalo manjše število končnih elementov.
Polinom višjega reda posledično zahteva vpeljavo dodatnih prostostnih stopenj. Zato bo v tej podpoglavju izpeljan nestandardni končni element za prečne pomike s šestimi prostostnimi stopnjami. Poleg prečnega pomika in zasuka na obeh koncih elementa se bosta vpeljala še dodatni prečni pomik in zasuk na sredini elementa.
Čeprav se spet izhaja iz Bernoulli–Eulerjeve hipoteze, je funkcija prečnih pomikov *v* (glede na lokalni koordinatni sistem $(x \in [0, L], y)$, sedaj izražena kot popolni polinom petega reda:

$$v = \alpha_1 + \alpha_2 \cdot x + \alpha_3 \cdot x^2 + \alpha_4 \cdot x^3 + \alpha_5 \cdot x^4 + \alpha_6 \cdot x^5$$

V *Mathematici* se tako definira ustrezni polinom:

■ Interpolacijske funkcije za specialni koncni element

```
In[1]:= Nf[x_] := α6 x^5 + α5 x^4 + α4 x^3 + α3 x^2 + α2 x + α1;
```

Konstante $\alpha_1, \alpha_2, \alpha_3, \alpha_4, \alpha_5, \alpha_6$ pa se določijo s pomočjo že znanih robnih pogojev ter dveh pogojev na sredini elementa:

```
In[2]:= resitve =
        Solve[{Nf[0] == Y1, Nf'[0] == ϕ1, Nf[L / 2] == Y2,
          Nf'[L / 2] == ϕ2, Nf[L] == Y3, Nf'[L] == ϕ3},
         {α1, α2, α3, α4, α5, α6}]
```

Zaradi podobnosti nekaterih simbolov *Mathematica* izpiše opozorilo o morebitnih napakah, ki pa ne vpliva na izračun (ker je program zaznal tri potencialne napake, izpiše tudi opozorilo, da med tem izračunom ne bo več opozarjal na to napako):

```
General::spelll :
 Possible spelling error: new symbol name "ϕ1" is
   similar to existing symbol "α1".

General::spelll :
 Possible spelling error: new symbol name "ϕ2" is
   similar to existing symbol "α2".

General::spelll :
 Possible spelling error: new symbol name "ϕ3" is
   similar to existing symbol "α3".

General::stop :
 Further output of General::spelll will be
   suppressed during this calculation.
```

Kljub opozorilom pa *Mathematica* vseeno tudi izpiše korektne rešitve:

Out[2]= $$\left\{\left\{\alpha 1 \to Y1,\ \alpha 3 \to -\frac{23\,Y1 - 16\,Y2 - 7\,Y3 + 6\,L\,\phi 1 + 8\,L\,\phi 2 + L\,\phi 3}{L^2},\right.\right.$$
$$\alpha 4 \to -\frac{-66\,Y1 + 32\,Y2 + 34\,Y3 - 13\,L\,\phi 1 - 32\,L\,\phi 2 - 5\,L\,\phi 3}{L^3},$$
$$\alpha 5 \to -\frac{4\,(17\,Y1 - 4\,Y2 - 13\,Y3 + 3\,L\,\phi 1 + 10\,L\,\phi 2 + 2\,L\,\phi 3)}{L^4},$$
$$\left.\left.\alpha 6 \to \frac{4\,(6\,Y1 - 6\,Y3 + L\,\phi 1 + 4\,L\,\phi 2 + L\,\phi 3)}{L^5},\ \alpha 2 \to \phi 1\right\}\right\}$$

in izračunane vrednosti je potrebno še dodeliti konstantam:

```
In[3]:= α1 = α1 /. resitve[[1]];
        α2 = α2 /. resitve[[1]];
        α3 = α3 /. resitve[[1]];
        α4 = α4 /. resitve[[1]];
        α5 = α5 /. resitve[[1]];
        α6 = α6 /. resitve[[1]];
```

Funkcija prečnega pomika oz. interpolacijska funkcija je tako:

```
In[9]:= ExpandAll[Nf[x]]
```

Out[9]=
$$Y1-\frac{23x^2\,Y1}{L^2}+\frac{66x^3\,Y1}{L^3}-\frac{68x^4\,Y1}{L^4}+\frac{24x^5\,Y1}{L^5}+\frac{16x^2\,Y2}{L^2}-\frac{32x^3\,Y2}{L^3}+\frac{16x^4\,Y2}{L^4}+\frac{7x^2\,Y3}{L^2}-\frac{34x^3\,Y3}{L^3}+\frac{52x^4\,Y3}{L^4}-\frac{24x^5\,Y3}{L^5}+x\,\phi1-\frac{6x^2\,\phi1}{L}+\frac{13x^3\,\phi1}{L^2}-\frac{12x^4\,\phi1}{L^3}+\frac{4x^5\,\phi1}{L^4}-\frac{8x^2\,\phi2}{L}+\frac{32x^3\,\phi2}{L^2}-\frac{40x^4\,\phi2}{L^3}+\frac{16x^5\,\phi2}{L^4}-\frac{x^2\,\phi3}{L}+\frac{5x^3\,\phi3}{L^2}-\frac{8x^4\,\phi3}{L^3}+\frac{4x^5\,\phi3}{L^4}$$

Za lažjo določitev vseh šestih interpolacijskih funkcij, ki pripadajo vsem šestim prostostnim stopnjam, se definira vektor, ki zajame vse elementu pripadajoče prostostne stopnje:

```
In[10]:= prostostnestopnje = {Y1, φ1, Y2, φ2, Y3, φ3};
```

Vektor interpolacijskih funkcij se sedaj izračuna tako, da se iz funkcije prečnega pomika zberejo posamezni členi, ki pripadajo posameznim prostostnim stopnjam:

```
In[11]:= Nvv = Table[Coefficient[Nf[x], prostostnestopnje[[i]]],
          {i, 6}]
```

Out[11]=
$$\left\{1-\frac{23x^2}{L^2}+\frac{66x^3}{L^3}-\frac{68x^4}{L^4}+\frac{24x^5}{L^5},\ x-\frac{6x^2}{L}+\frac{13x^3}{L^2}-\frac{12x^4}{L^3}+\frac{4x^5}{L^4},\ \frac{16x^2}{L^2}-\frac{32x^3}{L^3}+\frac{16x^4}{L^4},\ -\frac{8x^2}{L}+\frac{32x^3}{L^2}-\frac{40x^4}{L^3}+\frac{16x^5}{L^4},\ \frac{7x^2}{L^2}-\frac{34x^3}{L^3}+\frac{52x^4}{L^4}-\frac{24x^5}{L^5},\ -\frac{x^2}{L}+\frac{5x^3}{L^2}-\frac{8x^4}{L^3}+\frac{4x^5}{L^4}\right\}$$

Razvidno je, da se sedaj dobljene interpolacijske funkcije razlikuje od interpolacijskih funkcij za klasični oz. standardni končni element (glej prejšnje podpoglavje), kar velja tudi za funkcije, ki pripadajo prečnima pomikoma in zasukoma začetnemu oz. končnemu vozlišču.

Še bolj nazorno pa je ta razlika opazna pri izrisu vseh interpolacijskih funkcij. Funkcija, ki pripada prečnemu pomiku začetnega vozlišča, je tako:

```
In[12]:= Plot[Nvv[[1]] /. L → 1, {x, 0, 1}];
```



Pri osnovnem zapisu (torej brez dodatnih parametrov) ukaza Plot *Mathematica* poišče porazdelitev vrednosti koordinate in točke, ki ležijo primerno daleč od neke povprečne vrednosti, izpusti, saj smatra, da so takšne točke posledica morebitne singularnosti funkcije, ki se izrisuje.

Ker *Mathematica* vse točke izračuna, ne pa tudi izriše, se lahko zahteva ponovni prikaz zadnje slike, tokrat z izrecno zahtevo, da se prikažejo vse točke: *Show[%,PlotRange->All]*:

```
In[13]:= Show[%, PlotRange → All];
```



Opcija *PlotRange→All* bi se lahko uporabila tudi že pri ukazu *Plot* in zato se lahko pri izrisu funkcije, ki pripada zasuku začetnega vozlišča, že upošteva možnost, da bo del krivulje spet izpadel iz slike, in zato se že takoj zahteva izris celotne funkcije:

```
In[14]:= Plot[Nvv[[2]] /. L → 1, {x, 0, 1}, PlotRange → All];
```



Interpolacijski funkciji, ki pripadata sredinskemu vozlišču, pa se npr. izrišeta na eni sliki, saj je že na osnovi slike mogoče jasno določiti, kateri prostostni stopnji pripada posamezna krivulja:

```
In[15]:= Plot[{Nvv[[3]] /. L → 1, Nvv[[4]] /. L → 1}, {x, 0, 1},
          PlotRange → All];
```



Podobno se stori še za končno vozlišče:

```
In[16]:= Plot[{Nvv[[5]] /. L → 1, Nvv[[6]] /. L → 1}, {x, 0, 1},
           PlotRange → All];
```



S temi interpolacijskimi funkcijami je sedaj mogoče izpeljati masno in togostno matriko (kljub drugačnemu število prostostnih stopenj sta ukaza identično enaka kot pri izpeljavi standardnega končnega elementa):

■ matriki

```
In[17]:= K =
           Integrate[EI Outer[Times, D[Nvv, {x, 2}], D[Nvv, {x, 2}]],
            {x, 0, L}];
         M = Integrate[m Outer[Times, Nvv, Nvv], {x, 0, L}];
```

Za njun pregledni izpis v matrični obliki se lahko spet uporabi ukaz *MatrixFom[matrika]*. Togostna matrika je tako:

```
In[19]:= MatrixForm[K]
```

Out[19]//MatrixForm=

$$\begin{pmatrix} \frac{5092\,EI}{35\,L^3} & \frac{1138\,EI}{35\,L^2} & -\frac{512\,EI}{5\,L^3} & \frac{384\,EI}{7\,L^2} & -\frac{1508\,EI}{35\,L^3} & \frac{242\,EI}{35\,L^2} \\ \frac{1138\,EI}{35\,L^2} & \frac{332\,EI}{35\,L} & -\frac{128\,EI}{5\,L^2} & \frac{64\,EI}{7\,L} & -\frac{242\,EI}{35\,L^2} & \frac{38\,EI}{35\,L} \\ -\frac{512\,EI}{5\,L^3} & -\frac{128\,EI}{5\,L^2} & \frac{1024\,EI}{5\,L^3} & 0 & -\frac{512\,EI}{5\,L^3} & \frac{128\,EI}{5\,L^2} \\ \frac{384\,EI}{7\,L^2} & \frac{64\,EI}{7\,L} & 0 & \frac{256\,EI}{7\,L} & -\frac{384\,EI}{7\,L^2} & \frac{64\,EI}{7\,L} \\ -\frac{1508\,EI}{35\,L^3} & -\frac{242\,EI}{35\,L^2} & -\frac{512\,EI}{5\,L^3} & -\frac{384\,EI}{7\,L^2} & \frac{5092\,EI}{35\,L^3} & -\frac{1138\,EI}{35\,L^2} \\ \frac{242\,EI}{35\,L^2} & \frac{38\,EI}{35\,L} & \frac{128\,EI}{5\,L^2} & \frac{64\,EI}{7\,L} & -\frac{1138\,EI}{35\,L^2} & \frac{332\,EI}{35\,L} \end{pmatrix}$$

masna pa:

```
In[20]:= MatrixForm[M]
```

Out[20]//MatrixForm=

$$\begin{pmatrix} \frac{523\,L\,m}{3465} & \frac{19\,L^2\,m}{2310} & \frac{4\,L\,m}{63} & -\frac{8\,L^2\,m}{693} & \frac{131\,L\,m}{6930} & -\frac{29\,L^2\,m}{13860} \\ \frac{19\,L^2\,m}{2310} & \frac{2\,L^3\,m}{3465} & \frac{2\,L^2\,m}{315} & -\frac{L^3\,m}{1155} & \frac{29\,L^2\,m}{13860} & -\frac{L^3\,m}{4620} \\ \frac{4\,L\,m}{63} & \frac{2\,L^2\,m}{315} & \frac{128\,L\,m}{315} & 0 & \frac{4\,L\,m}{63} & -\frac{2\,L^2\,m}{315} \\ -\frac{8\,L^2\,m}{693} & -\frac{L^3\,m}{1155} & 0 & \frac{32\,L^3\,m}{3465} & \frac{8\,L^2\,m}{693} & -\frac{L^3\,m}{1155} \\ \frac{131\,L\,m}{6930} & \frac{29\,L^2\,m}{13860} & \frac{4\,L\,m}{63} & \frac{8\,L^2\,m}{693} & \frac{523\,L\,m}{3465} & -\frac{19\,L^2\,m}{2310} \\ -\frac{29\,L^2\,m}{13860} & -\frac{L^3\,m}{4620} & -\frac{2\,L^2\,m}{315} & -\frac{L^3\,m}{1155} & -\frac{19\,L^2\,m}{2310} & \frac{2\,L^3\,m}{3465} \end{pmatrix}$$

2. Modeliranje konstrukcije s končnimi elementi

Tako pridobljeni matriki se sedaj lahko uporabita za modeliranje konstrukcije (lahko tudi v kombinaciji z matrikama standardnega končnega elementa, če jo npr. potrebno ali želeno). V tem zgledu bo ponovno analizirana konzola iz vaje 10 (http://www.geocities.com/mcsDISK/e10.pdf) in sicer z enim samim novim končnim elementom, kar že v osnovni diskretizaciji pomeni štiri prostostne stopnje (kar je bilo prej doseženo šele z dvema končnima elementoma). Togostna in masna matrika konstrukcije bosta tako reda 4*4:

■ Konzola iz Vaje 10

■ diskretizacija z 1 koncnim elementom

```
In[21]:= Kk = Table[0, {i, 4}, {j, 4}];
         Mk = Table[0, {i, 4}, {j, 4}];
```

Če se smatra, da je element npr. na levem, začetnem koncu vpet, se navidezno prečrtajo prvi dve vrstici in prva dva stolpca, v togostno in masno matriko konstrukcije pa se tako prenesejo zgolj preostali členi. Za prenos podatkov je najprimerneje uporabiti ukaz *Do[operacija,{zunanji števec,začetna meja zunanjega števca, končna meja zunanjega števca},{notranji števec,začetna meja notranjega števca, končna meja notranjega števca}]* :

```
In[23]:= Do[Kk[[i, j]] = K[[i + 2, j + 2]], {j, 1, 4}, {i, 1, 4}]
         Do[Mk[[i, j]] = M[[i + 2, j + 2]], {j, 1, 4}, {i, 1, 4}]
```

Opomba: zadnje štiri vrstice bi lahko zapisali tudi kompaktnejše kot:

```
In[21]:= Kk = Table[K[[i + 2, j + 2]], {j, 1, 4}, {i, 1, 4}];
         Mk = Table[M[[i + 2, j + 2]], {j, 1, 4}, {i, 1, 4}];
```

3. Iskanje lastnih frekvenc

Spet se izhaja direktno iz osnovne oblike problema (torej iskanja determinante, ki zagotovi netrivialno rešitev homogenega sistema linearnih enačb), kar vodi do karakterističnega polinoma v obliki:

In[25]:= **Det[Kk − ω^2 Mk]**

$$\text{Out[25]=}\ \frac{3145728\,EI^4}{35\,L^8}-\frac{302252032\,EI^3\,m\,\omega^2}{40425\,L^4}+\frac{237568\,EI^2\,m^2\,\omega^4}{14175}-\frac{9387008\,EI\,L^4\,m^3\,\omega^6}{2406702375}+\frac{16\,L^8\,m^4\,\omega^8}{343814625}$$

katere rešitve so lastne krožne frekvence (izraz ni prikazan v celoti):

In[26]:= **Solve[% == 0, ω]**

$$\text{Out[26]=}\ \Bigg\{\Bigg\{\omega\to-\sqrt{\Bigg(-\frac{1}{2}\sqrt{\Bigg(-(402419808185876448\,EI^3)\Big/\Bigg(343\,L^{12}\sqrt{\Bigg(\frac{74286186496\,EI^2}{49\,L^8\,m^2}+(20192526336\,5^{2/3}\,EI^2)\Big/\Big(\big(198003490+9\,i\sqrt{50970136468945}\big)^{1/3}\,L^8\,m^2\Big)+\frac{1}{L^8\,m^2}\Big(98304\,\big(5\,\big(198003490+9\,i\sqrt{50970136468945}\big)\big)\,\text{^}\,(1/3)\,EI^2\Big)\Bigg)}\,m^3\Bigg)+\frac{148572372992\,EI^2}{49\,L^8\,m^2}-\frac{20192526336\,5^{2/3}\,EI^2}{\big(198003490+9\,i\sqrt{50970136468945}\big)^{1/3}\,L^8\,m^2}-\frac{1}{L^8\,m^2}\Big(98304\,\big(5\,\big(198003490+9\,i\sqrt{50970136468945}\big)\big)^{1/3}$$

oziroma nekoliko krajše (tudi ta rezultat ni prikazan v celoti):

In[27]:= **N[%]**

$$\text{Out[27]=}\ \left\{\left\{\omega \to -1.\sqrt{\left(-0.5\sqrt{\left(-\frac{1.17324\times10^{14}\,\mathrm{EI}^3}{L^{12}\sqrt{\frac{(1.7142\times10^{9}-1.86265\times10^{-9}\,i)\,\mathrm{EI}^2}{L^8\,m^2}}\;m^3}+\frac{(2.83393\times10^{9}+1.86265\times10^{-9}\,i)\,\mathrm{EI}^2}{L^8\,m^2}\right)}-0.5\sqrt{\frac{(1.7142\times10^{9}-1.86265\times10^{-9}\,i)\,\mathrm{EI}^2}{L^8\,m^2}}+\frac{20953.1\,\mathrm{EI}}{L^4\,m}\right)}\right\},\right.$$

Eliminirati je potrebno še nepravilne in zavajajoče imaginarne vrednosti, kar prinese nekaj napredka (tudi ta rezultat ni prikazan v celoti):

In[28]:= **Chop[%, 10^-8]**

$$\text{Out[28]=}\ \left\{\left\{\omega \to -1.\sqrt{\left(-0.5\sqrt{-\frac{2.8337\times10^{9}\,\mathrm{EI}^3}{L^{12}\sqrt{\frac{\mathrm{EI}^2}{L^8\,m^2}}\;m^3}+\frac{2.83393\times10^{9}\,\mathrm{EI}^2}{L^8\,m^2}}-20701.5\sqrt{\frac{\mathrm{EI}^2}{L^8\,m^2}}+\frac{20953.1\,\mathrm{EI}}{L^4\,m}\right)}\right\},\right.$$

in tako končno sledijo prve štiri lastne krožne frekvence:

```
In[29]:= PowerExpand[%]
```

$$\text{Out[29]}= \left\{\left\{\omega \to -\frac{3.51602\sqrt{EI}}{L^2\sqrt{m}}\right\}, \left\{\omega \to \frac{3.51602\sqrt{EI}}{L^2\sqrt{m}}\right\}, \left\{\omega \to -\frac{22.1578\sqrt{EI}}{L^2\sqrt{m}}\right\}, \left\{\omega \to \frac{22.1578\sqrt{EI}}{L^2\sqrt{m}}\right\}, \left\{\omega \to -\frac{63.3466\sqrt{EI}}{L^2\sqrt{m}}\right\}, \left\{\omega \to \frac{63.3466\sqrt{EI}}{L^2\sqrt{m}}\right\}, \left\{\omega \to -\frac{281.596\sqrt{EI}}{L^2\sqrt{m}}\right\}, \left\{\omega \to \frac{281.596\sqrt{EI}}{L^2\sqrt{m}}\right\}\right\}$$

Razvidno je, da sedaj dobljene lastne krožne frekvence predstavljajo boljši približek kot rezultati, dobljeni z diskretizacijo konzole z dvema standardnima končnima elementoma (ob enakem številu prostostnih stopenj), saj napake prve frekvence sploh ni več opaziti (glej numerične rešitve, podpoglavje II.5), za drugo lastno frekvenco napaka znaša manj kot 0.6 %, pri tretji frekvenci pa je tudi opazen napredek približka (napaka je padla s 17.9 % na sprejemljivih 2.6 %).

Prednost predstavljenega posebnega končnega elementa je torej očitna, saj je za dosego boljše konvergence ob istem številu prostostnih stopenj potrebnih manj končnih elementov in posledično manj računskega napora ob sestavljanju globalne togostne in masne matrike. Slabost tega končnega elementa je zgolj nekoliko daljša izpeljava obeh matrik, ki pa se tako ali tako izvede zgolj enkrat samkrat, nato pa so vse prednosti novega končnega elementa trajno na voljo.

NALOGE ZA SAMOSTOJNO DELO

1. Poišči lastne frekvence konzole dolžine L, diskretizirane z dvema posebnima končnima elementoma enakih dolžin

REŠITVE

$$\omega_1 = \frac{3.51602}{L^2}\cdot\sqrt{\frac{EI}{m}} \qquad \omega_5 = \frac{203.861}{L^2}\cdot\sqrt{\frac{EI}{m}}$$

$$\omega_2 = \frac{22.0346}{L^2}\cdot\sqrt{\frac{EI}{m}} \qquad \omega_6 = \frac{320.789}{L^2}\cdot\sqrt{\frac{EI}{m}}$$

$$\omega_3 = \frac{62.781}{L^2}\cdot\sqrt{\frac{EI}{m}} \qquad \omega_7 = \frac{685.988}{L^2}\cdot\sqrt{\frac{EI}{m}}$$

$$\omega_4 = \frac{122.586}{L^2}\cdot\sqrt{\frac{EI}{m}} \qquad \omega_8 = \frac{1351.1}{L^2}\cdot\sqrt{\frac{EI}{m}}$$